

# Camel99 Linker

October 23, 2021

Brian Fox, Kilworth Ontario Canada

Version 1.5 with binary program

Last revision June 19,2022



## Table of Contents

Introduction.....	4
Notes On Version 1.5.....	5
Preparation.....	6
Quick Instructions to use LINK.....	7
Using External Code.....	9
DEF MYPROG * the vector must be in the DEF to call it.....	10
MYPROG DATA WRKSP1,ENTRY * this is the vector.....	10
WKRSP1 BSS >20 * this is space for the registers.....	10
ENTRY MOV R1,R2 * the code starts here.....	10
RTWP.....	10
Call External Sub-routines.....	11
Using Strings in Assembly Language.....	15
Parameter Passing to a Sub-Program.....	17
Loading Parameters into External workspace.....	19
Scripting Link Jobs.....	20
Camel99 Workspace Registers and Usage.....	22
TI-99 Assembler Object Code Tag Meanings.....	23

# Introduction

The TI-99 Editor/Assembler (E/A) cartridge and the associated Assembler program gives us the ability to write programs in 9900 Assembly language. The program begins its life as a text file called “source code”. The source code is fed to the Assembler and the output of the Assembler process is called “object code”. Object code cannot be run by the CPU but must be first read and processed by a program called a LINKER. TI-99 has a “linker” built into the loader program in the E/A cartridge.

## Wait! Forth Does Not Need a Linker

Forth compiles source directly to memory and runs it. There are no intermediate steps. So why are we talking about a linker for Forth?

Sometimes it would be nice to use tested Assembly Language code that already exists rather than re-typing it in Forth Assembler which can lead to errors. The Camel99 Linker lets us load object code into Forth’s universe and run it like regular Forth words.

Other times we have Assembler code that we would like to test interactively. Forth’s interpreter with this linker is a great way to do that. We can dump memory, examine DATA statements, change registers and start code all from the Forth command line. We can even write short Forth program to exercise the Assembly language code.

This LINKER reads the object code file which contains special embedded information in the file that allows the machine code to be organized correctly in memory as well as allowing chunks of code from different programs to find each other’s components and sub-routines by name. This is accomplished by putting specific Assembly language labels into the object file with the actual code. The DEF and REF directives cause labels to be embedded in the object code file. Since those labels are real text, in the object file, it is relatively simple to read the file and make those labels “words” in the Forth dictionary.

The standard linker for TI-99 Object code is the E/A cartridge menu item three (3). In larger computers the LINKER processes some object code files and generates a new file called an “executable” file. The TI-99 E/A cartridge has what is more rightly called a “linking-loader” because it takes a number of object files and links them directly into memory and from there the program can be run by typing in the name of the program into the E/A cartridge menu.

## Linking Order Matters

This linker is a bit more like Forth in that it does not allow forward references. What this means is that if you REF some code in a program before the corresponding DEF is linked into Forth you will get a not recognized error from Forth. This is because the linker can only resolve a REF that is already in the DEF word list.

It is slightly less flexible than allowing REFs to any label and then reporting unresolved REFs after the fact, but it was a design decision for simplicity. If you really want forward REF support contact me directly at [brian.fox@brianfox.ca](mailto:brian.fox@brianfox.ca) or @theBF on atariage.com

## Notes On Version 1.5

This version of LINKER99 does not require you to compile the linker to use it.

A new program file called LINKEXE is provided that is a binary program that loads with Editor/Assembler menu option five. (RUN Program File)

The entire Camel99 Forth system plus the linker and the DIR utility are part of this executable file.

It's much faster to begin using Linker99.

## Save Your Program

After linking object files in Forth memory you can now also save a memory image back to disk with the SAVE command. You must be sure to use the BOOTS command to set the name of the DEF label that will start the program.

# Preparation

Since this linker sits on top of Forth and Forth is the actual command line interpreter, some familiarity with Camel99 Forth is essential to using this linker. Refer to the Camel99 Forth manual as you see fit.

## Things to keep in mind:

- All commands and numbers that you type must be separated by one or more space characters
- All arguments for commands are taken from a structure called the data stack
- The interpreter will try to lookup any **word** you give it:
  - If the **word** is in the “dictionary” Forth will run the code connected to that **word**.
  - If the word is not found it tries to convert it to a number
  - If it IS a number the number is put onto the DATA stack. (nothing more)
  - If number conversion fails, you get an error message.

Since arguments go onto the DATA stack by default this is where Forth words get their arguments. This means that **arguments are typed in first**, before the command, so they fall onto the data stack.

Commands are typed after the arguments so they can remove their arguments from the data stack. This may seem backwards but it simplifies the interpreter (takes less space) and allows commands to be strung together without the use of data in variables which is powerful once you get the hang of it.

## Handy Forth Commands to Know

**DECIMAL**    Change the interpreter to use base 10 numbers

**HEX**            Change the interpreter to use base 16 numbers (hexadecimal)

**.**                Print the number on the top of the stack in current base and remove from stack

**U.**            Print the top of the stack in current base as unsigned and remove from stack

**\**                Line comment. Ignore the rest of the line. (needs a space after it)

**@**    (Fetch)    Replace the address on the top of the data stack with the contents of the address.

**!**    (Store)    Move the value from the 2<sup>nd</sup> stack item into the address that is the 1<sup>st</sup> stack item.

**PAGE**            Clears the screen and puts the cursor at top left


**COLD**            Restarts Forth with a clean dictionary

## Example:

```
HEX                \ Switch to hexadecimal
99 E000 !          \ Places HEX 99 into memory address HEX E000
E000 @ .          \ Fetch value at E000 and print it.
```

## Quick Instructions to use LINK

1. Assemble your assembly language program as you normally would to generate an object file but follow these rules:



- a. Be sure your Assembly language program includes at least one DEF statement. This DEF name will be brought into Forth's dictionary by the Camel99 Linker and is the way that Forth locates your program in memory.
- b. **Do not use the AORG directive in your program.** This will allow Forth to put the program in memory that is safe to use from Forth's point of view. (Low RAM at >2000 to >3FFF)
- c. **\*The binary image of your program cannot exceed 8K in this version**

## Starting LINKEXE

- Place CAMEL99 disk in DSK1. Or mount ask DSK1 in your emulator.
- Place LINKER99 disk in DSK2. Or mount as DSK2 in your emulator.
- Insert Editor/Assembler Cartridge into the TI-99 console
- Select Editor/Assembler cartridge and press menu option 5 (RUN PROGRAM FILE)
- At the prompt type: DSK2.LINKEXE
- When the program loads you will see a help screen.

## Linking Your Object Files

Type **NEW** to initialize the system for new object files and remove all DEFs from the DEF dictionary word list.

Link all your required object files with: `LINK DSK4.MYFILE` (example)

You can link all the files you need to link at this point.

LINK will display all the DEF labels it found in your object code when it is done.

Also when LINK finishes, it puts the DEFS wordlist first in the dictionary search order so all the DEF labels in your code will be visible to Forth.

Type: `WORDS` and you will see all the labels that you put in DEF statements in your Assembly language programs

### What does it all mean?

After linking an object file, when you invoke the name of a DEF from your Assembly Language Program it puts the address of that DEF on the top of the Forth data stack. That address is the actual address in memory where the linker put the code associate with that DEF label.

If you know what that DEF is you can invoke a Forth command that will take that address and do the right thing with it.



## Using External Code

*\* It is important that you know what the code associated with your code labels actually does, so that you do the right thing with them inside Forth. Forth has no way of knowing if a label points to code or data, sub-routines or sub-programs.*

## DATA

**DATA, BSS, BYTE** directives will return their start address to the Forth data stack. You can treat them like Forth variables. You can fetch the contents with the fetch (@) operator and the dot (.) operator will print a value that is on the DATA stack. You can also use ? So see the contents of an address if you loaded DSK1.TOOLS first. Store integers with the store (!) operator.

```
HEX 1234 MYDATA !    \ Put >1234 into memory location MYDATA
MYDATA @ .          \ fetches the data from MYDATA & prints it
MYDATA ?            \ fetch and print the contents at MYDATA (needs INCLUDE DSK1.TOOLS)
```

You can also declare data as words in the Forth side of the dictionary if you want to with the command **EXTERN:** (This is not very useful but it's here if you need it.)

```
MYDATA EXTERN: INT99    \ mydata becomes a Forth word
```

## Sub-Routines

Sub-routines are code blocks that end with **B \*R11** or the **RT** pseudo-instruction.

These can be run from Forth with the **CALL** command. They will return to Forth when they complete.

```
MYSUB CALL           \ BL to MYSUB and return to Forth on proper completion.
```

A faster way to call sub-routines in a program is to declare them as external sub-routines with the **EXT-SUB:** directive. In Forth you declare them in **interpreter mode** like a constant or a variable. (\*\*Do not put them inside a colon definition\*\*)

```
MYSUB EXT-SUB: COOLTHING
```

This assembles a symbolic BL statement rather than calling through Forth's R4. (TOS register) By using **EXT-SUB: COOLTHING** is created as Forth word that knows how to start itself.

This example compiles the code **BL @COOLTHING** internally so it just does the right thing.

## Sub-Programs

Sub-programs are code blocks that have a vector data structure. A “vector” begins with the workspace address followed by the entry address. In your Assembly language program it would look like this:

```
DEF  MYPROG          * the vector must be in the DEF to call it

MYPROG  DATA  WRKSP1,ENTRY  * this is the vector

WKRSP1   BSS >20          * this is space for the registers

ENTRY    MOV R1,R2          * the code starts here
        More code...

        RTWP

        END
```

Sub-programs **MUST BE** called with the **BLWP** command. They will return to Forth when they complete if the program ends with **RTWP**.

### Example:

```
MYPROG BLWP
```

The line of code will branch to workspace at the vector MYPROG, run the code at the vector’s entry address and return to Forth.

A faster way to call a sub-program is to declare an EXT-PROG: to create a Forth word that does a symbolic BLWP call to the external code.

```
MYPROG EXT-SUB: PROG1
```

If that all made sense to you then can begin now. Examples code follows for more clarity.

## Call External Sub-routines

The program below shows us a simple down-counting sub-routine that we can CALL from Forth.

Notice there is no AORG statement. AORG would cause the Assembler to force the program to reside at a specific place in memory. The Forth system assumes it has all memory from HEX A000 to FFFF. By not using the AORG directive the program can be “relocated” by our LINK command to the low RAM 8K memory block at HEX 2000.

---

```
* Exampl,A99  for Camel99 LINK

*****
* You must have at least one DEF statement *
* The text after DEF will appear in Forth  *
* after you run LINK                       *
*****

* This program will use Forth's workspace.
* R0..R3 R5 & R8 are free to use.

      DEF SUB1          * Allow SUB1 to be used by external programs

      LI  R1,>FFFF      * Load R1 with 65,536

SUB1   DEC R1           * decrement R1
      JNE SUB1          * if R1<>0 GOTO SUB1
      RT               * return from sub-routine

      END
```

---

### Link & Run the Exampl Program

We have already assembled the EXAMPL1,A99 program and produced the file EXAMP1,OBJ

Place your Camel99 diskette in DSK1.

Place the Linker diskette in another DSK of your choice. Our examples use DSK4.

- **If you have NOT already done so do:**
  - Start CAMEL99 from E/A menu option 5 by typing: DSK1.CAMEL99
  - Compile the LINKER with: INCLUDE DSK4.LINKER
- When the linker is compiled type: LINK DSK4. EXAMP1,OBJ
- When that completes run the program with: SUB1 CALL <enter>

After a second the ‘ok’ prompt will mean that you are back in Forth.

This means the 65536 iteration loop ran and completed successfully and returned to Forth

## CLS Sub-routine Example

Link the program DSK4.CLS,OBJ and you should see that you have two DEF sub-routines available. SUB1 and CLS as shown below.

```
ok
LINK DSK4.EXAMP1,OBJ
Linking DSK4.EXAMP1,OBJ
99/4 AS

DEFS
SUB1
1 words
Elapsed time =0:00.23 ok
LINK DSK4.CLS,OBJ
Linking DSK4.CLS,OBJ
99/4 AS

DEFS
CLS SUB1
2 words
Elapsed time =0:00.43 ok
```

CLS CALL will now clear the VDP screen using external program seen below. You can even put them together now as a FORTH word and run them sequentially.

Consider the following Forth code:

```
: TEST
  CLS CALL
  CR ." Running 65,536 iterations..."
  SUB1 CALL
;
```

Just like that you have woven two Assembly language programs into a Forth word.

---

```

* Demonstration program by @mathew180
* modified to be sub-routine for CAMEL99 LINKER 2021 FOX

```

```

DEF CLS          * Clear the screen in 40 column mode

```

```

*
* VDP Memory Map
*

```

```

VDP RD EQU >8800      * VDP read data
VDP STA EQU >8802      * VDP status
VDP WD EQU >8C00      * VDP write data
VDP WA EQU >8C02      * VDP set read/write address

```

```

* Workspace

```

```

WRKSP EQU >8300      * Workspace is shared with Forth
R0LB EQU WRKSP+1     * R0 low byte req'd for VDP routines

```

```

* Program execution starts here

```

```

CLS      LIMB 0
         CLR  R0          * Set the VDP address to zero
         MOVB @R0LB,@VDPWA * Send low byte of VDP RAM write address
         ORI  R0,>4000     * Set read/write bits 14 and 15 to write (01)
         MOVB R0,@VDPWA   * Send high byte of VDP RAM write address

         LI   R1,>2000     * Set high byte to 32 (>20)
         LI   R2,>3C0      * bytes in the 40 column screen
LOOP1    MOVB R1,@VDPWD   * Write byte to VDP RAM
         DEC  R2
         JNE  LOOP1
         RT           * return to caller
         END

```

## How it Works

When we LINK the file DSK4.EXAMP1,OBJ Forth opens the file and begins reading the file line by line. Here is the actual file contents. Notice it is text data.

```
0000A          A0000B0201BFFFFB0601B16FEB045B7F70EF          0001
50004SUB1  7FD75F          0002
:          99/4 AS          0003
```

Before every piece of data in this file there is a hexadecimal number between 0 and F. These are called tags and each TAG has a meaning for example

- zero(0) means the following data is the program ID
- ‘B’ means put this following number into the next available memory location and advance the memory pointer
- ‘.’ means the following data is a text string that is the name of the Assembler that created this object file

It is beyond the scope of this manual to explain the entire system. For our purposes we only need to know that linker knows what all the tags mean and it does the right thing with the data following each tag. If you want to learn more about tags see page 309 of the TI-99 Editor/Assembler manual. Or look at the Forth source code in the file DSK2.LINKER.

An important tag for our purposes is ‘5’. You can see it at the beginning of line 2 in the object file. This tag says the following data is the “relocatable” entry address of some code and the text following the entry address is the actual DEF label name from the Assembly language program.

The linker puts that text label into the DEFS vocabulary in Forth and computes the “real” address in memory where that piece of code has been placed by the linker.

At the Forth command line when you type the name of that “DEF”, SUB1 in this case, the address is automatically placed onto the Forth data stack.

When we type the command CALL it picks up the address from the Forth data stack and does a BL instruction (branch and link) to that address in the top of DATA stack cache register. (R4)

The code runs as a sub-routine and if the Assembly language programmer has ended the code with the RT instruction or B \*R11, the program will return and it will then return to Forth.

# Using Strings in Assembly Language

Below is a simple program that has no code but only data. The data is in the form of “counted strings”. These are strings of text where the first byte is the length of the string. This format is commonly used in PASCAL and is something of a standard way to store strings in Forth memory as well.

Notice the EVEN directive is used to make sure that the program address is an “even” number after each string even if the string+length ends on an odd numbered address. TMS9900 doesn’t like odd-numbered addresses.

---

```
* example2 Counted strings in Assembler
```

```
DEF MSG1
DEF MSG2
DEF MSG3
DEF MSG4
```

```
MSG1  BYTE 16
      TEXT 'Now is the time '
      EVEN
```

```
MSG2  BYTE 17
      TEXT 'for all good men '
      EVEN
```

```
MSG3  BYTE 19
      TEXT 'to come to the aid '
      EVEN
```

```
MSG4  BYTE 17
      TEXT 'of their country.'
      EVEN
```

```
END
```

---

We have assembled this program producing the file EXAMP2,OBJ

We can LINK DSK4.EXAMP2,OBJ and see that MSG1, MSG2, MSG3 and MSG4 are visible in the DEFS vocabulary in Forth. We can print these strings very easily using Forth’s COUNT and TYPE commands. COUNT converts the address of a counted string into an address where the text begins and a length value. After COUNT executes both numbers will be on the DATA stack.

TYPE takes two input arguments, an address and a length. That’s exactly what COUNT provides!

So we can print an Assembler string in Forth like this:

MSG1 COUNT TYPE and we will see : Now is the time

### **If you had a lot of strings to check...**

If we had a large number of strings we would be better off to make new Forth command to print these strings by simply combining COUNT TYPE in a word definition. It is this easy:

```
: PRINT    COUNT TYPE ;  
( defines PRINT that takes a counted string argument)
```

Now we can do this from the Forth command line:

```
MSG1 PRINT  
MSG2 PRINT  
MSG3 PRINT  
MSG4 PRINT
```

Each string in our Assembly language program will be printed on the screen just as if they were created in Forth.



## Parameter Passing to a Sub-Program

In the code EXAMP3,A99 below we create a “sub-program” called PROG1. When we say sub-program we mean code that uses its own separate workspace and must be called with the BLWP instruction.

To pass a parameter to PROG1 we use a feature of the TMS9900 that occurs when BLWP is used. After the BLWP instruction is invoked the machine switches to another workspace. It remembers how to get back by putting the original workspace in R13 of the new workspace. With this we can use R13 to refer back to any register in the caller’s workspace by using indexed addressing.

With a BLWP we leave the Forth workspace and enter a new one. If we placed the number HEX FFFF on the Forth data stack, we can retrieve it in a DEF sub-program with indexed addressing.

To get a register from the caller program use:

$$(R\# * 2) + \text{old\_workspace} = \text{address of the register in caller's workspace}$$

You can see the instruction at label ENTRY is doing this to load a value from R4 in Forth workspace to R0 of the new workspace. R4 is the cache register for the top of the data stack in Camel99 Forth.

### IMPORTANT DETAIL TO REMEMBER

The Forth LINKER has the command BLWP that let’s you call a sub-program however...

BLWP is a simple command that takes the vector to call from the Forth top-of-stack register. So if try to start PROG1 with this command line:

```
HEX FFFF PROG1 BLWP
```

We will **not** get the FFFF parameter because it is the second item on the stack. Instead we will pass the address of PROG1 to the sub-program. So how can we get FFFF into the top of stack?

We must declare an “external” sub-program with line like this:

```
PROG1 EXT-PROG: TESTPROG1
```

The EXT-PROG: directive assembles the instruction BLWP @PROG1, bypassing the Forth stack entirely. With this we can invoke TESTPROG1 and FFFF will be passed to PROG1 and the loop will count down from that value and return -1 on the Forth DATA stack.

```
HEX FFFF TESTPROG1 .
```

Notice at the end of the program, just before RTWP (return to workspace) we do the reverse and put the contents of R0 into R4 in the Forth workspace and return to Forth.

```

* Examp3,A99  for Camel99 LINK

*****
* GET/PUT a parameter to/from Forth Top of Stack *
* with an external sub-program                  *
*****

*****
* Usage: From LINKEXE
*       LINK DSK2.EXAMP3,OBJ
*       45  PROG1 BLWP .  * (use DOT to see the result)
*
*       DEF XWKSP          * Export the local workspace address
*       DEF PROG1          * export the sub-program vector

XWKSP    BSS 32            * external program workspace registers
PROG1    DATA XWKSP,ENTRY * BLWP to this vector from Forth

****  PROGRAM STARTS HERE  ****
ENTRY   MOV @8(R13),R0     * move Forth's R4 to local R0

* Do a loop and Fill some other registers for the demo
LBL1     MOV R0,R2
          MOV R2,R3
          MOV R3,R4
          MOV R4,R5
          MOV R5,R6
          MOV R6,R7
          DEC R0
          JOC LBL1         * GOTO LBL1 until R0<0

          MOV R0,@8(R13)   * Update Forth's top of stack register
          RTWP             * return from sub-program to Forth

END

```

## Loading Parameters into External workspace

We might want to put parameters into the registers of our Assembly language program in order to test specific functions that use parameters in registers. There are couple of way to accomplish this in Forth but the simplest I can think of is below.

### Access External Workspace Registers

A nice feature of the 9900 CPU for this purpose is that the registers are in memory. Since program EXAMP3,OBJ has a DEF for the workspace we can get at the registers from Forth with @ and !. We can also make use of the word CELLS which computes the number of memory cells from any input argument.

```
\ Loading external XWKSP registers in Forth
HEX
BEEF XWKSP !          \ put BEEF in R0 of XWKSP
DEAD XWKSP 1 CELLS + ! \ Put DEAD in R1 of XWKSP
ABBA XWKSP 8 CELLS + ! \ Put ABBA in R8 of XWKSP
XWKSP 8 CELLS + @ .    \ Fetch and print the contents of XWKSP R8
```

### Create Blocks of Parameters as Forth DATA

Let's say we need to feed 6 parameters into the Assembly language program's workspace. We can create blocks of data with parameters inside as shown below. (You don't need to type the comments)

*\*Notice the space between every item in the code. Forth delimits each 'word' by a space. The comma is not a delimiter. It is an actual sub-routine that writes a number into the next available memory cell.*

```
HEX \ Switch interpreter to base 16
\ Make some data arrays in memory with the register data we need
\ Register #   R0      R1      R2      R3      R4      R5
CREATE PARMS1 00FF , 0000 , 001A , DEAD , BEEF , ABBA ,
CREATE PARMS2 0001 , 000D , 000A , BEEF , DEAD , BAAB ,
```

Using the previous demo program we see our workspace is called XWKSP . To copy PARMS1 into our the external workspace we use the CMOVE command which is moves characters (bytes) .

CMOVE needs three input arguments ( src-addr dst-addr length -- )

```
\ at the Forth command line type
DECIMAL
PARMS1 XWKSP 6 CELLS CMOVE    \ This copies 12 bytes from PARMS1 TO XWKSP
```

If we ran the sub-program called PROG1 after copying these parameters into XWKSP the program would start and have its registers pre-loaded. Whatever we put in R4 will be the loop counter.

# Scripting Link Jobs

Since we have an interpreter available it becomes possible to write a script that automates the job of linking a bunch of object files and even creating the external declarations.

To read the script file use the INCLUDE command:

```
INCLUDE DSK4.LINKSCRIPT
```

You can now run PROG1 PROG2 and GO.

**\*WARNING\*** If you run GO you must reset the machine to exit. It never returns.

The file example follows:

```
CR .( Camel199 LINKER TEST SCRIPT Sept 1 2021)

LINK DSK4.THING1
LINK DSK4.THWACK

\ *** make external declarations ***

\ The next line is ANS Forth. Don't panic.
ONLY FORTH DEFINITIONS
\ It means "only" search the FORTH word list
\ and any new definitions will go in the Forth word list

\ The next line puts the DEFS word list "also" in the interpreter
\ search order so we can find those names as well.
ALSO DEFS

THINK EXT-PROG: PROG1 \ BLWP to THINK. Increments VAR1
THWACK EXT-PROG: PROG2 \ BLWP to THWACK. Display chars

LOOP EXTERN: GO \ runs THINK and THWACK forever
```

# **APPENDIX**

# Camel99 Workspace Registers and Usage

If you are running sub-routines that share the Forth workspace be aware of the register usage below.

If you need all the registers your programs should declare their own workspace and vectors so you can BLWP into your sub-program from Forth and safely return to Forth with RTWP.

Forth default workspace is: >8300

- R0    scratch register. External Sub-routines can use it.
- R1    scratch register. External Sub-routines can use it.
- R2    scratch register. External Sub-routines can use it.
- R3    scratch register. External Sub-routines can use it.
- R4    Forth TOP of stack cache. Can be used with care
- R5    scratch register. External Sub-routines can use it.
- R6    parameter stack pointer. DO NOT USE
- R7    return stack pointer        DO NOT USE
- R8    Forth 'W' register        External Sub-routines can use it.
- R9    Forth IP (Interpreter pointer) DO NOT USE
- R10   Forth's "NEXT" routine cache DO NOT USE
- R11   9900 sub-routine linkage register - ONLY for sub-routines
- R12   9900 CRU register                                - OR - scratch register
- R13   Forth Multi-tasker LINK to next task        - OR - scratch register
- R14   Forth Multi-tasker Program counter        - OR - scratch register
- R15   Forth Multi-tasker task Status register    - OR - scratch register

## TI-99 Assembler Object Code Tag Meanings

Object format tags are created in object code when your assembly language code is assembled. The tags provide the Loader with the information it needs to load the object code.

CODE	Name	Actions
<b>0</b>	Module ID	Sets PROGNAME string in the linker
<b>1</b>	Absolute Entry	Not supported. Aborts with an error
<b>2</b>	Relocatable Entry	Not supported. Aborts with an error
<b>3</b>	Relocatable External REF	REF is relocated and stored in the corresponding DEF. Error is DEF not found
<b>4</b>	Absolute External REF	REF value is stored in the corresponding DEF. Error is DEF not found
<b>5</b>	Relocatable External DEF	Relocated value is stored as the DEF name
<b>6</b>	Absolute External DEF	Absolute value is stored as the DEF name
<b>7</b>	Record checksum	Ignored
<b>8</b>	Ignored Checksum	Ignored
<b>9</b>	Absolute Load Address	Linker address pointer is set to the absolute value in object file
<b>A</b>	Relocatable Load Address	Linker address pointer set to the Relocated address in object file
<b>B</b>	Absolute DATA	Absolute value of data is compiled into the program image
<b>C</b>	Relocatable DATA	Relocated value of data is compile into the program image
<b>D</b>	Load BIAS	Ignored
<b>E</b>	Undefined	Ignored
<b>F</b>	End of record	Ignores remainder of record
<b>G</b>	undefined	Ignored
<b>H</b>	undefined	Ignored
<b>I</b>	Program Segment ID	Ignored
<b>:</b>	End of File	Display version of the Assembler that created the object file.